

**ESERCITAZIONI DEL CORSO DI
PROGETTAZIONE ASSISTITA DA COMPUTER
CLM ING. VEICOLI
Rev.2019 (da slides Prof.Leonardo Bertini)**


Organizzazione di un modello ad EF

- Definizione dei nodi
- Definizione degli elementi (tipo e connessione)
- Proprietà del materiale
- Vincoli
- Carichi

Struttura del programma ANSYS Parametric Design Language APDL/ ANSYS Classic

Ambiente di base (BEGIN)

- Gestione files
- Comandi di sistema




- Definizione “mesh”
- Prop. materiale
- Vincoli
- Carichi

- Vincoli
- Carichi
- Opzioni soluzione
- Soluzione

- Analisi risultati
 - grafica
 - tabulare
- Comb. carico

Tipica sessione di lavoro ANSYS

Ambiente di base (BEGIN)



Pre-processor (PREP7)

Preparazione modello


Solutore (SOLU)

Soluzione

Post-processor (POST1, POST26)

Analisi risultati

Modalità di lavoro in ANSYS



- esecuzione immediata comandi
- preferibile per piccoli modelli/apprendimento

- esecuzione comandi in sequenza durante la lettura del file
- preferibile per grossi modelli
- parametrizzazione

Uso di file Script di comandi

Files di comandi: files di testo contenenti i comandi di ANSYS
Possono essere scritti con un qualunque *editor* in formato ASCII
(Es. Notepad)

/INPUT, *Fname*,*txt*

IN ALTERNATIVA:

- Eseguire manualmente i comandi attraverso i *menu* di dialogo.
- Introdurre manualmente i comandi attraverso il *prompt* dei comandi.

SUCCESSIVAMENTE:

E' possibile ritrovare i comandi impartiti nel file *.log* o file di registro di ANSYS.

Spesso questo modo di procedere è molto comodo per sviluppare il modello, ottenendo il file di Input, non conoscendo la sintassi.

Vantaggi del file Script di comandi

- Possibilità di intervenire sul file per correggere il modello
- Parametrizzazione: i valori numerici nei comandi possono essere sostituiti da variabili su cui è possibile operare con un linguaggio di programmazione (calcoli, operazioni logiche, cicli, subroutines, etc.); in tal modo diviene possibile usare lo stesso file per ottenere:
 - ✓ mesh parametriche per una geometria data
 - ✓ geometrie parametriche in grado di rappresentare intere famiglie di componenti

STRUTTURA COMANDI ANSYS

I comandi in ANSYS sono una stringa alfanumerica (il comando stesso), seguita da una serie di parametri separati da virgola, di cui alcuni sono valori numerici (o espressi come variabile) mentre altri sono parole chiave del comando.

È fondamentale rispettare l'ordine della sequenza dei parametri.

COMANDO, par. 1, par.2, par. 3, par. 4,



DENOMINAZIONE
COMANDO

- *APDL non è Case Sensitive, quindi tutti i comandi e le variabili possono essere scritti indifferentemente in MAIUSCOLO o in minuscolo*
- *i comandi possono anche essere abbreviati purché univocamente individuabili. Es.: FINISH → FINI*

COMANDI DI USO GENERALE

/PREP7
/SOLU
/POST1
/POST 26



INGRESSO NEI DIVERSI AMBIENTI
DI LAVORO

FINISH

USCITA DAGLI AMBIENTI DI LAVORO

/TITLE, titolo in caratteri alfanumerici

ATTRIBUISCE UN TITOLO AL MODELLO

Creazione modello in ANSYS

Manuale/diretta



Introduzione diretta
nodi ed elementi

- utile per apprendimento e per interventi di correzione sul modello
- comandi che facilitano la definizione di “mesh” regolari

Modellatore solido



Definizione della geometria
del pezzo e sua suddivisione
automatica in elementi

- utile per geometrie complesse
- minore controllo sulla disposizione degli elementi


Sistemi di riferimento in ANSYS/1

Il programma utilizza diversi SR, ciascuno dei quali ha una funzione specifica. In particolare è necessario conoscere il SR secondo vengono interpretate le grandezze in Ingresso/Uscita


- Sistemi GLOBALI
- Sistemi LOCALI
- Sistemi NODALI
- Sistemi di ELEMENTO

SISTEMI GLOBALI (GLOBAL COORDINATE SYSTEMS)


- 3 SR predefiniti aventi la stessa origine ed identificati da un numero



CARTESIANO (0)



CILINDRICO (1)




SFERICO (2)

USO:

- Introduzione coordinate nodali e parametri geometrici

SISTEMI LOCALI (LOCAL COORDINATE SYSTEMS)

- SR definiti dall'utente, che può controllare:
 - origine
 - tipo (cartesiano, cilindrico, sferico)
 - rotazione assi rispetto ai SR globali




USO:

- Introduzione coordinate nodali e parametri geometrici

SISTEMI NODALI (NODAL COORDINATE SYSTEMS)

- SR cartesiani aventi origine nel nodo di appartenenza
- per “default” gli assi sono paralleli a quelli del SR cartesiano globale
- possono essere ruotati (comando NROTAT) al fine di introdurre carichi o vincoli secondo direzioni inclinate





USO:

- Introduzione vincoli e carichi concentrati

SIST. DI ELEMENTO (ELEMENT COORDINATE SYSTEMS)

- SR cartesiani definiti per ogni elemento
- dipendono da:
 - tipo elemento
 - orientazione elemento
 - scelte utente



N.B.: l'asse x_E è sempre quello che congiunge i primi due nodi: i e j

USO:

- Introduzione prop. materiale e carichi distribuiti

INSERIMENTO NODI/1

Inserimento nodo singolo

N, n° nodo, X, Y, Z o (R, θ , Z) (R, θ , ϕ)

Esempio:

N,1,10,10

N,5,20,10

Inserimento nodi aggiuntivi tra nodi esistenti, interpolando le coordinate che definiscono la posizione

FILL, nodo iniziale, nodo finale

Esempio:

FILL,1,5

INSERIMENTO NODI/2

Visualizzazione coordinate nodi

NLIST, nodo iniziale, nodo finale

Esempio:

NLIST,1,5

NLIST,2,4

Cancellazione nodi

NDELE, nodo iniziale, nodo finale

NDELE,ALL

Esempio:

NDELE,1,2


NDELE,ALL

INSERIMENTO NODI/3

Generazione insiemi di nodi (con distribuzione lineare)::

NGEN, n° ins., incr., nodo in., nodo fin., passo, Δx , Δy , Δz , Rapp.

```
N, 1,  
N, 4, 35  
FILL,1,4  
NGEN, 5, 4, 1, 4,,0,10
```




INSERIMENTO NODI/4

Generazione insiemi di nodi (con distribuzione progressiva):

NGEN, n° ins., incr., nodo in., nodo fin., passo, Δx , Δy , Δz , Rapp.

NDELE, 5, 20

NGEN, 5, 4, 1, 4,,0,10,0,2



SISTEMI DI RIFERIMENTO/1

Utilizzo di un sistema di riferimento cilindrico:

CSYS, n° SR

CSYS,1


N,1,10

N,10,10, 90

FILL,1,10

CSYS,0

FILL,1,10



SISTEMI DI RIFERIMENTO/2

Creazione sistema di riferimento locale:

LOCAL, n° SR, tipo, X_0 , Y_0 , Z_0

N,1,0,10


N,10,20,10

FILL,1,10

LOCAL,11,1,20,0,0

N,20,10,0

FILL,10,20



INTRODUZIONE ELEMENTI/1

Per ogni elemento è disponibile una scheda che ne illustra le caratteristiche

Definizione tipi di elemento da usare :


ET, n° id., n° libreria, Keyopt 1, Keyopt 2,....

Libreria locale *Libreria ANSYS*

ET,1,182 ! (plane stress)

ET,1,182,,,1 !(assialsimmetrico)

Figure 182.1: PLANE182 Geometry



INTRODUZIONE ELEMENTI/2

Introduzione elemento:

E, nodo I, nodo J, nodo K, nodo L,....

....
nodi

....
E,1,2,6,5

Generazione elementi:



EGEN, n° ins., incr.nod., el. in., el. fin.

EGEN,3,1,1

EGEN,4,4,1,3

Cancellazione(lista) elementi:

EDELE (ELIST), elem. in., elem. fin.



PROPRIETÀ MATERIALE

Introduzione proprietà materiale:

MP, proprietà, n° mat., valore 1, valore 2, valore 3,....

MP,EX,1,210000

MP,NUXY,1,0.3


*N.B.: in APDL non sono implementate le unità di misura.
L'utente deve prestabilire un sistema di unità e rimanere
coerente quando introduce grandezze fisiche derivate.
Es. lunghezze in [mm], forze in [N] → tensioni [MPa]*

SELEZIONE NODI/ELEMENTI

E' possibile rendere attiva solo una parte del modello. I comandi con ALL si applicano alla sola parte attiva.

Selezione nodi:

NSEL, tipo selez., criterio, sottocriterio, valore min., valore max.



NSEL, ,LOC,Y,-0.01,0.001

Selezione elementi:

ESEL, tipo selez., criterio,

Nota: quasi tutti i comandi che riguardano i nodi iniziano per N, gli elementi per E

VINCOLI

Introduzione vincoli:

UX, UY, UZ,
ROTX; ROTY, ROTZ,
ALL

D, n° nodo, g.d.l., valore

D,1,UX, 0


NSEL,,LOC,Y,-0.1,0.001

D,ALL,UY,0

NSEL,ALL

Cancellazione vincoli:

DDELE, n° nodo, g.d.l.



CARICHI/1

Introduzione carichi concentrati:

F, n° nodo, g.d.l., valore


F,18,FY,10

F,19,FX,-10

Cancellazione carichi concentrati:

FDELE, n° nodo, g.d.l.

FX, FY, FZ,
MX, MY, MZ,



CARICHI/2



Introduzione carichi distribuiti:

SF, ALL, PRES, valore

```
NSEL,,LOC,Y,39.99,41  
SF,ALL,PRES,-10  
NSEL,ALL
```

Verso positivo
specificato nella scheda
dell'elemento
(in questo caso entrante)

Figure 182.1: PLANE182 Geometry



SOLUZIONE

...

FINISH !Conclude qualunque operazione precedente

/SOLU !Entra nel solutore

SOLVE !Risolve

/POST1 ! Entra nel post-processore



PLDISP,1

PLNSOL,S,Y

```

/clear,all
/PREP7
CSYS,1
N,1,10
N,9,10,90
FILL,1,9
NGEN,9,9,1,9,,20/9
N,82,30
CSYS,0
N,86,30,30
N,90,0,30
FILL,82,86
FILL,86,90
NGEN,15,5,86,90,,0,60/14
ET,1,182
MP,EX,1,205000
MP,PRXY,1,0.3
E,1,2,11,10
EGEN,8,1,1
EGEN,9,9,1,8
E,86,87,92,91
EGEN,4,1,-1  !(*)
EGEN,14,5,-4
NSEL,,LOC,Y,-1,0.001
D,ALL,UY,0
NSEL,,LOC,X,29.99,31
D,ALL,UX,0
NSEL,,LOC,Y,89.99,91
SF,ALL,PRES,-10
ALLS
FINISH


```



* If *IEL1* is negative, *IEL2* and *IEINC* are ignored and the last $|IEL1|$ elements (in sequence backward from the maximum element number) are used as the pattern to be repeated.

MODELLAZIONE SOLIDA

Definizione della geometria
del pezzo



Suddivisione automatica in
elementi

DEFINIZIONE GEOMETRIA/1

ELEMENTI COSTITUTIVI DI UN MODELLO SOLIDO

- PUNTI (KEYPOINTS)
- LINEE (LINES)
- SUPERFICI (AREAS)
- VOLUMI (VOLUMES)


PROCEDURE POSSIBILI:

- BOTTOM-UP
- TOP-DOWN

DEFINIZIONE GEOMETRIA/2



BOTTOM – UP: il modello si basa su costituenti elementari che vengono introdotti in ordine di complessità

Keypoints





DEFINIZIONE GEOMETRIA/2

BOTTOM – UP: il modello si basa su costituenti elementari che vengono introdotti in ordine di complessità





DEFINIZIONE GEOMETRIA/2

BOTTOM – UP: il modello si basa su costituenti elementari che vengono introdotti in ordine di complessità



DEFINIZIONE GEOMETRIA/2


BOTTOM – UP: il modello si basa su costituenti elementari che vengono introdotti in ordine di complessità



Adatto per ogni tipo di geometria. Può risultare più laborioso dell'altro per geometrie semplici.

DEFINIZIONE GEOMETRIA/3



TOP-DOWN: il modello si basa su “primitive” (forme geometriche semplici) di superficie e volume



Primitiva



DEFINIZIONE GEOMETRIA/3

TOP-DOWN: il modello si basa su “primitive” (forme geometriche semplici) di superficie e volume





DEFINIZIONE GEOMETRIA/3

TOP-DOWN: il modello si basa su “primitive” (forme geometriche semplici) di superficie e volume



DEFINIZIONE GEOMETRIA/3

TOP-DOWN: il modello si basa su “primitive” (forme geometriche semplici) di superficie e volume



Adatto soprattutto per geometrie semplici, in cui è possibile individuare facilmente le primitive costituenti. Nella pratica è comunque frequente il ricorso ad approcci misti

GESTIONE ELEMENTI MODELLO

Agli elementi del modello solido si applicano comandi simili a quelli impiegabili per nodi ed elementi

KDELE, LDELE, ADELE, VDELE

Cancellazione keypoints (KPs), linee, aree e volumi

KLIST, LLIST, ALIST, VLIST

Elenco KPs, linee, aree e volumi

KSEL, LSEL, ASEL, VSEL

Selezione KPs, linee, aree e volumi

METODO BOTTOM-UP

Introduzione Keypoints

K, n° Keypoint, X, Y, Z

K,1

K,2,0,10

K,20,10,0




METODO BOTTOM-UP

Introduzione automatica Keypoints

KFILL, KP1, KP2, ...

K,1
K,2,0,10
K,20,10,0
KFILL,2,20




METODO BOTTOM-UP

Introduzione automatica Keypoints : effetto del SR

KFILL, KP1, KP2, ...

```
K,1  
K,2,0,10  
K,20,10,0  
CSYS,1  
KFILL,2,20
```



INSERIMENTO AUTOMATICO KPS

Generazione insiemi di KPs:

KGEN, n° ins., KP in., KP fin., passo, Δx , Δy , Δz , incr.

KDELE,ALL


CSYS,0

K, 1,

K, 4, 35

KFILL,1,4

KGEN, 5, 1, 4,,0,10,0,4



INSERIMENTO KEYPOINTS

Visualizzazione coordinate Keypoints

KLIST, nodo iniziale, nodo finale

Esempio:

KLIST,1,5

KLIST,2,4

KLIST

Cancellazione Keypoints

KDELE, nodo iniziale, nodo finale

KDELE,ALL

Esempio:

KDELE,1,2

KDELE,ALL

SISTEMI DI RIFERIMENTO LOCALI

Creazione sistema di riferimento locale:

LOCAL, n° SR, tipo, X_o , Y_o , Z_o

KDELE,ALL

CSYS,0

K,1,0,10


K,10,20,10

KFILL,1,10

LOCAL,11,1,20,0,0

K,20,10,0

KFILL,10,20



Introduzione Linee

L, P1, P2, ...

Tra keypoints

(Crea KPs)

K,1

CSYS,1


K,2,10,90

K,5,10,0

(Crea linee)

L,1,2

L,1,5



Introduzione Linee


LARC,P1,P2,PC,RAD

Arco di circonferenza

↖ *KP defining plane of arc and center of curvature side*

(Usa i KPs precedenti)
KFILL,2,5

LARC,4,5,1,5
LARC,3,4,1,4
LARC,2,3,1,3




Introduzione Linee

LFILLT,L1,L2,RAD,..

(Usa i KPs
precedenti)

```
LDELE,ALL  
KDELE,6,999  
L,1,2,  
L,1,5  
LFILLT,1,2,4
```

Raccordo tra due linee con un
estremo in comune



Introduzione Linee

SPLINE, KP1, KP2, ..., KP6, XV1, YV1, ZV1, XV2, YV2, ZV2

2-6 punti

slope inizio e fine spline

“Spline” cubica tra KPs

Percorso elicoidale

LDELE, ALL

KDELE, ALL

CSYS, 1

K, 1, 50


KGEN, 21, 1, , , 0, 360/20, 50/20

SPLINE, 1, 2, 3, 4, 5, 6

SPLINE, 6, 7, 8, 9, 10, 11

SPLINE, 11, 12, 13, 14, 15, 16

SPLINE, 16, 17, 18, 19, 20, 21



Introduzione Linee

BSPLIN

genera una cubica di “best fit” su dei KPs

LTAN

genera una linea tangente ad una data

LANG

genera una linea ad un angolo dato con una esistente

LDRAG

genera linee facendo scorrere dei KPs lungo un percorso

LCOMB

unisce due linee

LXTEND

prolunga una linea

Introduzione Aree

A,P1,P2,...,P18

K,1


K,2,10

K,3,10,20

K,4,0,20,20

A,1,2,3,4

genera un'area attraverso i vertici (lati rettilinei, anche non complanare)



Introduzione Aree

A,P1,P2,...,P18

genera un'area attraverso i vertici (lati rettilinei, anche non complanare)

K,1

K,2,10

K,3,10,20


K,4,0,20,20

A,1,2,3,4

ADELE,1

A,1,2,3

A,2,3,4



Introduzione Aree

AL,L1,L2,...,L10

genera un'area attraverso i lati

ADELE,ALL

LDELE,ALL

L,1,2

L,1,3


L,1,4

L,2,4

L,3,4

AL,1,3,4

AL,2,3,5



Introduzione Aree

AL,L1,L2,...,L10

genera un'area attraverso i lati

ADELE,ALL

LDELE,ALL

L,1,2

L,1,3

L,1,4

L,2,4


L,3,4

AL,1,3,4

AL,2,3,5

ADELE,ALL

AL,1,2,4,5



Introduzione Aree

ADRAG,L1,L2,...,L6,LP1,LP2,...LP6

genera aree facendo scorrere linee su un percorso

Filetto trapezio

K,1,50

CSYS,1

KGEN,6,1, , ,0,360/5,50/5

SPLINE,1,2,3,4,5,6,

K,7,50,0,10


K,8,35,0,15

K,9,35,0,-5

L,1,7

L,7,8

L,1,9



Introduzione Aree

ADRAG,L1,L2,...,L6,LP1,LP2,...LP6

genera aree facendo scorrere linee su un percorso (linee)

Filetto trapezio

K,1,50

CSYS,1

KGEN,6,1, , ,0,360/5,50/5

SPLINE,1,2,3,4,5,6,

K,7,50,0,10

K,8,35,0,15


K,9,35,0,-5

L,1,7

L,7,8


L,1,9

ADRAG,6,7,8, , , ,1,2,3,4,5



Introduzione Aree

Osservazione: per una buona riuscita dell'operazione di "drag" è opportuno che gli oggetti da far scorrere giacciono sul piano ortogonale alla linea di scorrimento, tracciato nel suo punto iniziale; a tal fine è possibile controllare le pendenze iniziale e finale della linea stessa.



Introduzione Aree

AROTAT,L1,L2,...,L6,PAX1,PAX2,ANG

genera aree facendo ruotare linee attorno ad un asse

Anello a sezione circolare

CSYS,0

K,1,50

K,2,50,10


K,3,

K,4,,,50

CIRCLE,1,10,2,,360

*CIRCLE, PCENT, RAD,
PAXIS, PZERO, ARC, NSEG*

cerchio in piano // xz
asse // y



Introduzione Aree

AROTAT,L1,L2,...,L6,PAX1,PAX2,ANG

linee da ruotare punti asse di rot.

genera aree facendo ruotare linee attorno ad un asse

Anello a sezione circolare

```
CSYS,0
```

```
K,1,50
```


```
K,2,50,10
```

```
K,3,
```

```
K,4,,,50
```

```
CIRCLE,1,10,2,,360
```

```
AROTAT,1,2,3,4,,,3,4,360
```



Introduzione Aree

AFILLT,A1,A2,RAD

genera un raccordo tra due aree

```
CSYS,0
```

```
K,1,
```

```
K,2,50
```

```
K,3,50,50
```


```
K,4,0,50
```

```
K,5,,,50
```

```
K,6,,50,50
```

```
A,1,2,3,4
```

```
A,1,4,6,5
```



Introduzione Aree

AFILLT,A1,A2,RAD

genera un raccordo tra due aree

```
CSYS,0
```

```
K,1,
```

```
K,2,50
```

```
K,3,50,50
```

```
K,4,0,50
```


```
K,5,,,50
```

```
K,6,,50,50
```

```
A,1,2,3,4
```

```
A,1,4,6,5
```

```
AFILLT,1,2,20
```




Introduzione Aree

AOFFST,A1,DIST,...

genera una nuova area traslata lungo la normale all'area corrente

```
K,1, !origine  
K,2,50 !2° pt asse  
K,3,,50 !1° pt arco  
K,4,0,0,50 !2° pt arco  
LARC,3,4,1,50 !arco  
L,1,2 !asse  
ADRAG,1,,,,,2 !trascin  
!arco lungo asse  
AOFFST,1,20
```




Introduzione Volumi

V,P1,P2,...,P8

genera un volume tramite i vertici

```
K,1,  
K,2,50  
K,3,50,50  
K,4,0,50  
KGEN,2,1,4,,0,0,100  
V,1,2,3,4,5,6,7,8
```



Introduzione Volumi

V,P1,P2,...,P8

genera un volume tramite i vertici

K,1,

K,2,50

K,3,50,50


K,4,0,50

KGEN,2,1,4,,0,0,100

V,1,2,3,4,5,6,7,8

VDELE,1

V,1,2,3,4,5,6,6,5



Introduzione Volumi


V,P1,P2,...,P8

genera un volume tramite i vertici

```
K,1,  
K,2,50  
K,3,50,50  
K,4,0,50  
KGEN,2,1,4,,0,0,100  
V,1,2,3,4,5,6,7,8
```

```
VDELE,1  
V,1,2,3,4,5,6,6,5
```

```
VDELE,1  
V,1,2,3,4,5,5,5,5
```



Introduzione Volumi

$VA, A1, A2, \dots, A10$

genera un volume tramite le superfici laterali

Piramide a base quadrata

K,1,

K,2,50

K,3,50,50

K,4,0,50

K,5,25,25,50

A,1,2,3,4


A,1,2,5

A,2,3,5

A,3,4,5

A,4,1,5

VA,1,2,3,4,5




Introduzione Volumi

VDRAG,A1,A2,...,A6,L1,L2,...,L6

genera un volume facendo scorrere superfici su un percorso

Spira di molla

```
K,1,50  
CSYS,1  
KGEN,6,1,1,,0,360/5,50/5  
SPLINE,1,2,3,4,5,6,0,-1,0,0,1,0  
CSYS,0  
K,7,50,10  
CIRCLE,1,5,7  
AL,6,7,8,9  
VDRAG,1,,,,,1,2,3,4,5
```




Introduzione Volumi

VROTAT genera un volume ruotando superfici attorno a un asse

VOFFST genera un volume traslando aree lungo la loro normale (estrusione)


VEXT genera un volume traslando e scalando aree lungo un vettore definito tramite componenti (estrusione con espansione); interpola linearmente tra la forma iniziale e quella finale



VOFFST, NAREA, DIST, KINC

VEXT, NA1, NA2, NINC, DX, DY, DZ, RX, RY, RZ

METODO TOP-DOWN



Vantaggioso per corpi di forma semplice, in cui è facile riconoscere le primitive

Working planes

Le primitive sono definite con riferimento al “working plane” corrente. Inizialmente il WP coincide col piano “xy” del SR cartesiano globale. E’ possibile creare nuovi WP personalizzati.


WPLANE, WN, XOR, YOR, ZOR, XAX, YAX, ZAX, XPL, YPL, ZPL

genera un
WP per 3
punti

Origine

Punto su
asse “x”

3° punto sul
WP



Working planes

Altre istruzioni per definire i WPs

WPOFFS,xoff,yoff,zoff

Applica un «offset» all'origine del WP

WPROTA,thxy,thyz,thzx

Ruota il WP

WPCSYS,win,kcn

Allinea il WP con il SR kcn

Primitive di area

RECTNG,X1,X2,Y1,Y2


genera un rettangolo sul WP

```
RECTNG,0,10,0,20
```

```
WPLANE,,0,0,0,0,1,0,0,0,1
```

!oriento asse X come Y, Y come Z

```
RECTNG,0,10,0,20
```



WPLANE,WN,XOR,YOR,ZOR,XAX,YAX,ZAX,XPL,YPL,ZPL


Primitive di area

CYL4,XC,YC,RAD1,ANG1,RAD2,ANG2,(PROF)

genera un settore circolare (cilindro) sul WP

CYL4,5,5,15,90,5,270

!da 90° a 270°



Primitive di area

RPR4,NSIDES,XC,YC,RAD,ANG,(PROF)


genera un poligono regolare (prisma) sul WP

RPR4,6,10,5,20

!6 lati, raggio 20 circ. circoscritta

!ang per orientare poligono rispetto

!sistema rif. WP




Primitive di volume

BLOCK, X1, X2, Y1, Y2, Z1, Z2

genera un parallelepipedo sul WP

BLOCK,0,12.5,0,25,0,50

!coord. spigoli




Primitive di volume

CYL4,XC,YC,RAD1,ANG1,RAD2,ANG2,PROF

genera un settore cilindrico sul WP

CYL4,5,5,15,0,5,360,50




Primitive di volume

RPR4,NSIDES,XC,YC,RAD,ANG,PROF

genera un prisma regolare sul WP

RPR4,5,10,10,20,45,100



Primitive di volume

SPHERE


genera una sfera o un settore sferico

CONE

genera un cono o un tronco di cono

TORUS


genera un toro o un settore torico



Operazioni Booleane

BOPTN,KEEP,YES(NO)

Mantiene o no le entità originali,
accanto a quelle risultato
dell'operazione, default no



Operazioni Booleane : intersezione

Creano una nuova entità che rappresenta la parte comune di due entità esistenti. Le entità possono essere dello stesso livello o di livelli diversi

LINL, LINA, LINV

linea con linea, area o volume

AINA, AINV

area con area o volume

VINV

volume con volume

Operazioni Booleane : intersezione

Intersezione di pari livello

CYL4,,,10,,5,,50


WPLANE,,25,0,25,25,0,50,

25,25,25

CYL4,,,8,,4,,50

VINV,1,2


→wO rispetto a
origine base cilindro
maggiore, wx//z



WPLANE,WN,XOR,YOR,ZOR,XAX,YAX,ZAX,XPL,YPL,ZPL

Operazioni Booleane : intersezione

Intersezione di livello diverso
CYL4,,,10,,5,,50
WPLANE,,0,0,25,25,0,25,
0,25,25
RECTNG,-25,25,-25,25
AINV,7,1



Operazioni Booleane : somma


Creano una nuova entità che rappresenta l'unione di due entità esistenti. Le entità devono essere dello stesso livello

AADD **somma aree**

VADD **somma volumi**


Operazioni Booleane : somma

```
RECTNG,0,10,0,20  
CYL4,10,10,5  
AADD,1,2
```



Operazioni Booleane : somma

```
BLOCK,-20,20,-20,20,15,35  
RPR4,5,0,0,10,45,50  
VADD,1,2
```



Operazioni Booleane : overlap

Crea nuove entità di forma semplice che coprono l'intero spazio occupato da due entità esistenti.


LOVLAP **overlap linee**

AOVLAP **overlap aree**

VOVLAP **overlap volumi**


Operazioni Booleane : overlap

RECTNG,0,10,0,20
CYL4,10,10,5
AOVLAP,1,2



Operazioni Booleane : overlap

BLOCK,-20,20,-20,20,15,35
RPR4,5,0,0,10,45,50
VOVLAP,1,2



Operazioni Booleane : sottrazione

Creano una nuova entità sottraendo da un'entità data la parte comune con un'altra entità. Le entità possono essere dello stesso livello o di livelli diversi

LSBL, LSBA, LSBV

linea, area o volume da linea

ASBA, ASBL, ASBV

area, linea o volume da area

VSBV, VSBA

volume o area da volume

Operazioni Booleane : sottrazione


RECTNG,0,10,0,20

CYL4,10,10,5

ASBA,1,2




sottrae 2 da 1



Operazioni Booleane : sottrazione

```
BLOCK,-20,20,-20,20,15,35  
RPR4,5,0,0,10,45,50  
VSBV,1,2
```

↑
sottrae 2 da 1



Operazioni Booleane : sottrazione

Sottrazione di livello diverso

```
CYL4,,,10,,,,50
```


```
WPLANE,,0,0,25,25,0,25,  
0,25,25
```

```
RECTNG,-25,25,-25,25
```

```
ASBV,5,1
```



sottrae volume 1 (a cui sono
associate 4 aree) da area 5



Operazioni Booleane : Glue (incollaggio)

Mette in comune la zona di contatto tra due entità che si toccano (superfici tra volumi, linee tra superfici).

LGLUE linee


AGLUE aree

VGLUE volumi

GLUE \neq ADD: le entità rimangono separate con interfaccia comune

Operazioni Booleane : Glue (incollaggio)





```
BLOCK,-20,20,-20,20,15,35  
RPR4,5,0,0,10,45,15  
VGLUE,1,2
```





Prima dell'operazione di "glue" i due volumi non hanno superfici comuni


Dopo l'operazione di "glue" i due volumi, rimangono distinti (\neq VADD) ma hanno una nuova superficie comune

Operazioni Booleane: Glue (incollaggio)



ESEMPIO





ESEMPIO

```
/PREP7
```

```
RECTANGLE,0,30,0,30  
CYL4,0,0,0,0,10,360  
ASBA,1,2  
RECTNG,0,30,30,90  
AGLUE,ALL
```

Modello solido

```
LSEL,,LOC,Y,0,0  
DL,ALL,,UY  
LSEL,,LOC,X,0,0  
DL,ALL,,UX
```

Vincoli

```
SMRTSIZE,2
```

Suddivisione

```
LSEL,,LOC,Y,90,90  
SFL,ALL,PRES,-10  
ALLS
```

Carichi

```
MP,EX,1,210000
```

Materiale

```
ET,1,182  
AMESH,ALL
```


Mesh

```
/SOLU  
SOLVE
```



Soluzione

```
/POST1  
PLNSOL,S,Y
```

Post-processing



ESEMPIO



Files usati da ANSYS/1

File usati da ANSYS (file di *input*, file.log, file.db / .dbb)

Files di registro: è possibile recuperare la sequenza dei comandi impartiti ad ANSYS attraverso il file di registro (*.log*) che si trova nella cartella di lavoro

All'avvio di ANSYS è possibile impostare la cartella di lavoro e il *nome* di *default*(File) del Data Base (*.db*)

Per cambiare cartella di lavoro: **/CWD, DIRPATH**

Es.:/CWD,'D:\Files_ANSYS'

Per cambiare *nome* del Data Base: **/FILENAME,Fname**

Es.:/FILENAME,'LastraIntagliata'

Files usati da ANSYS/2

Impostato con *FILNAME* il jobname (*Fname*), ANSYS genera una serie di altri file:

Fname.err(file di testo contiene tutti gli errori)

Fname.rst(file di risultati)

Tutti questi file possono essere rimossi al termine della sessione di lavoro di ANSYS

E' inoltre possibile impostare un *titolo*, diverso dal jobname

Eseguendo il comando: **/TITLE**, *Title*

Es.:/TITLE,'Esercizio, Lastra con Intaglio'

Generalmente il titolo (che compare sugli output) è più informativo del semplice nome del File (*Fname*)

Files usati da ANSYS/3

Eseguendo un salvataggio:

File -> Save as Jobname.db

viene creato il file di DataBase *Fname.db* e la versione precedente viene archiviata come *Fname.dbb*

Nel file *.db* vengono immagazzinate tutte le informazioni del modello, in particolare:

- entità geometriche

- nodi e elementi

- soluzione se eseguita


Tuttavia si perde la sequenza di comandi. Viene salvato soltanto lo *stato* attuale del modello.

Chiudendo una sessione di lavoro di ANSYS, successivamente è possibile recuperare il file *.db* mediante il comando:

File -> Resume Jobname.db

SUDDIVISIONE IN ELEMENTI (Mesh)

Preliminarmente è necessario specificare i tipi di elementi desiderati ed attivarli insieme alle proprietà materiali, real constants, etc.



Controlli sul processo di suddivisione (1)

MSHKEY, PAR

PAR = 0 Free


= 1 Mapped

= 2 Mapped se possibile, altrimenti Free


MESHPE, PAR

PAR = 0 quadrilateri (2D) o esaedri (6facce) (3D)

= 1 triangoli (2D) o tetraedri (4 facce) (3D)



MSHKEY,0




MSHKEY,1

Controlli sul processo di suddivisione (2)


ESIZE, SIZE, NDIV

Specifica la grandezza (nelle unità di misura in uso) del lato dell'elemento sulle linee di confine di aree e volumi (SIZE) o il numero di suddivisioni di tali linee (NDIV)


SIZE=0.3



SIZE=0.2




SIZE=0.1




Controlli sul processo di suddivisione (3)

Sono possibili controlli locali del “mesh” attraverso comandi come **LESIZE** che consente di specificare la dimensione dei lati su singole linee.



Dimensioni generali = 0.2
(ESIZE=0.2)

Dimensioni su
questa linea = 0.05







Controlli sul processo di suddivisione (4): Free mesher avanzato (Smart size)

E' possibile richiamare un meshatore free di tipo avanzato con il comando:

SMRTSIZE, SIZLVL,

Numerosi altri parametri di controllo

Da 10 (mesh grossolano) a 1 (mesh fine)



Controlli sul processo di suddivisione (5): rifiniture locali

E' possibile raffinare la mesh localmente tramite i comandi:

NREFINE, Nin, Nfin, Npasso, livello, profondità...

(costruisce nei dintorni dei nodi interessati una nuova mesh più fine della precedente)

EREFINE, Ein, Efin, Epasso, livello, profondità...

(costruisce nei dintorni degli elementi interessati una nuova mesh più fine della precedente)

Generazione della “mesh”

Una volta fissate le opzioni, è possibile ottenere la suddivisione in elementi con i comandi:

AMESH, Ain, Afin, Apasso

Suddivisione di aree

VMESH, Vin, Vfin, Vpasso

Suddivisione di volumi

(quando la geometria del volume è particolarmente complessa, può risultare comodo prima operare le divisioni sulle linee, poi meshare le aree, e solo successivamente meshare nel volume)

Gli enti geometrici si “puliscono” dagli elementi con i comandi:

LCLEAR, Lin, Lfin, Lpasso

Pulitura delle linee

ACLEAR, Ain, Afin, Apasso


Pulitura delle aree

VCLEAR, Vin, Vfin, Vpasso

Pulitura dei volumi

Esempi di “mesh” di semplici solidi piani (free e mapped)

```
/PREP7  
CYL4,0,0,10,0,20,45  
ET,1,182
```




Esempi di “mesh” di semplici solidi piani (free e mapped)

```
/PREP7  
CYL4,0,0,10,0,20,45  
ET,1,182
```

(A) Mesh mapped, quadrilateri, 5
su ogni bordo:

```
ESIZE,,5  
MSHKEY,1  
MSHAPE,0  
AMESH,1
```



Esempi di “mesh” di semplici solidi piani (free e mapped)


```
/PREP7  
CYL4,0,0,10,0,20,45  
ET,1,182
```

(A) Mesh mapped, quadrilateri, 5
su ogni bordo:

```
ESIZE,,5  
MSHAPE,0  
MSHKEY,1  
AMESH,1
```

(B) Mesh free, triangoli di dim. 1
su ogni bordo:

```
ESIZE,1  
MSHKEY,0  
MSHAPE,1  
AMESH,1
```



Esempi di “mesh” di semplici solidi piani (free e mapped)

```
/PREP7  
CYL4,0,0,10,0,20,45  
ET,1,182
```

(A) Mesh mapped, quadrilateri, 5
su ogni bordo:


```
ESIZE,,5  
MSHAPE,0  
MSHKEY,1  
AMESH,1
```

(B) Mesh free, triangoli di dim. 1
su ogni bordo:

```
ESIZE,1  
MSHAPE,1  
MSHKEY,0  
AMESH,1
```


(C) Mesh free, quadrilateri di dim.
2 su ogni bordo:

```
ESIZE,2  
MSHKEY,0  
MSHAPE,0  
AMESH,1
```



Esempi di “mesh” di semplici solidi 3D (free e mapped)

```
/PREP7  
CYL4,0,0,10,0,20,45  
K,,0,0  
K,,0,10  
VROTAT,1,,,,,,,,5,6,45  
ET,1,185
```




Esempi di “mesh” di semplici solidi 3D (free e mapped)

```
/PREP7  
CYL4,0,0,10,0,20,45  
K,,0,0  
K,,0,10  
VROTAT,1,,,,,,,,5,6,45  
ET,1,185
```

(A) Mesh mapped, esadri, 5 su ogni bordo:

```
ESIZE,,5  
MSHAPE,0  
MSHKEY,1  
VMESH,1
```



Esempi di “mesh” di semplici solidi 3D (free e mapped)


```
/PREP7  
CYL4,0,0,10,0,20,45  
K,,0,0  
K,,0,10  
VROTAT,1,,,,,,,,5,6,45  
ET,1,185
```

(A) Mesh mapped, esaedri, 5 su ogni bordo:

```
ESIZE,,5  
MSHAPE,0  
MSHKEY,1  
VMESH,1
```

(B) Mesh free, tetraedri, di dim. 2 su ogni bordo:

```
ESIZE,2  
MSHKEY,0  
MSHAPE,1  
LESIZE,3,1  
VMESH,1
```



COMANDI DI SELEZIONE/1

E' possibile rendere attiva solo una parte del modello. I comandi con ALL si applicano alla sola parte attiva.

Selezione KPs

Range di valori per la selezione

KSEL, tipo selez., criterio, sottocriterio, valore min., valore max.

- S seleziona dall'intero modello (default)
- R seleziona dalla parte attiva
- ALL seleziona tutto
-

Dipende dal criterio:
Per KP = nessuno
Per LOC = X, Y, Z (1°, 2° o 3° coordinata)

- KP : numero KPs
- LOC: coordinate KPs
- ...

KSEL, ,LOC,Y,-0.01,0.001

COMANDI DI SELEZIONE/2

Esistono comandi di selezione per tutti i componenti del modello

NSEL : selezione nodi

ESEL : selezione elementi

LSEL : selezione linee

ASEL : selezione aree

VSEL : selezione volumi

La selezione delle linee in base alle coordinate si basa sul punto medio, mentre quelle di aree e volumi sul baricentro.

Esistono anche comandi di selezione «incrociati». Es.:

NSLE : seleziona i nodi connessi agli elementi attivi

ESLN : selezione gli elementi connessi ai nodi attivi

etc.

Vincoli

I vincoli possono essere applicati al modello solido e trasferiti al modello EF. Si possono anche applicare direttamente ai nodi.

D, N1, LABEL,VAL,...

vincolo su un nodo

DK, P1, LABEL,VAL,...

vincolo su un KP

DL, L1, A1,LABEL,VAL,...

vincolo su una linea

DA, L1, A1,LABEL,VAL,...

vincolo su una superficie

DTRAN

trasferisce i vincoli dal modello solido al modello EF(il trasferimento è cmq automatico da parte del solutore)

Vincoli di dipendenza

I vincoli di dipendenza si possono applicare direttamente ai nodi.

CP, ... vincola i gdl di due nodi ad avere lo stesso spostamento

CE,... vincola i gdl di due nodi a rispettare una condizione in forma di equazione algebrica lineare

CERIG,... vincola un gruppo di nodi a comportarsi come un insieme rigido

Carichi concentrati

I carichi concentrati possono essere applicati direttamente sui nodi del modello EF o sui KPs del modello solido per poi trasferirli al modello EF

F, N1, LABEL,VAL,...

carico su un nodo

FK, P1, LABEL,VAL,...

carico su un KP

FLIST, FKLIST

FDELE, FKDELE

elenco e cancellazione

FTRAN

trasferisce i carichi concentrati dal modello solido al modello EF (il trasferimento è cmq automatico da parte del solutore)

Carichi distribuiti

I carichi distribuiti possono essere applicati su linee e superfici. Si possono inoltre applicare alla mesh utilizzando nodi ed elementi

SF, ...

carico distribuito su nodi

SFE, ...

carico distribuito su elementi

SFL, ...

carico distribuito su una linea

SFA, ...

carico distribuito su una superficie

SFBEAM, ...

carico distribuito su travi

SFGRAD, ...

imposta un gradiente di carico

SF*LIST

elenco

SF*DELE

cancellazione

SFTRAN

trasferisce i carichi distribuiti dal modello solido al modello EF (il trasferimento è cmq automatico da parte del solutore)

Carichi di volume

I carichi distribuiti possono essere applicati su linee e superfici. Si possono inoltre applicare al mesh utilizzando nodi ed elementi

BF, ...	carico distribuito su nodi
BFE, ...	carico distribuito su elementi
BFL, ...	carico distribuito su una linea
BFA, ...	carico distribuito su una superficie
BF*LIST	elenco
BF*DELE	cancellazione
BFTRAN	trasferisce i carichi distribuiti dal modello solido al modello EF (il trasferimento è cmq automatico da parte del solutore)

Carichi inerziali

I carichi distribuiti inerziali simulano la presenza di campi di accelerazione, originanti forze di inerzia sul corpo

ACEL, ...


accelerazione X, Y o Z
(permette, ad esempio, di simulare la forza peso)

OMEGA, ...


velocità angolare (forze centrifughe)

DOMEGA, ...

accelerazione angolare



LINGUAGGIO PARAMETRICO/0




problema di partenza

LINGUAGGIO PARAMETRICO/1

PARAMETRIZZAZIONE DIMENSIONI

```
C*****
C*** RESET DATABASE
C*****
FINISH
/CLEAR
C***
C*** PARAMETRI
C***
L=100      ! Lunghezza
B=25       ! Larghezza
RI=12.5    ! Raggio foro
/PREP7
C***
C*** MODELLO SOLIDO
C***
RECTANGLE,0,B,0,B
CYL4,0,0,0,0,RI,360
ASBA,1,2
RECTNG,0,B,B,L
AGLUE,ALL
C***
C*** VINCOLI
LSEL,,LOC,Y,0,0
DL,ALL,,UY
LSEL,,LOC,X,0,0
DL,ALL,,UX
```



```

C*****
C*** RESET DATABASE
C*****
FINISH
/CLEAR
C***
C*** PARAMETRI
C***
L=100      ! Lunghezza
B=25       ! Larghezza
RI=12.5    ! Raggio foro
NE0=4    ! Numeri di elementi
NE1=5
NE2=8
/PREP7
C***
C*** MODELLO SOLIDO
C***
RECTANGLE,0,B,0,B
CYL4,0,0,0,0,RI,360
ASBA,1,2
RECTNG,0,B,B,L
AGLUE,ALL
C***
C*** VINCOLI
LSEL,,LOC,Y,0,0
DL,ALL,,UY
LSEL,,LOC,X,B,B
DL,ALL,,UX

```


```

C***
C*** SUDDIVISIONE
C***
LSEL,,LOC,Y,0,0
LESIZE,ALL,,NE0
LSEL,,LOC,Y,RI,B
LSEL,R,LOC,X,0,0
LESIZE,ALL,,NE0
CSYS,1
LSEL,,LOC,X,RI,RI
LESIZE,ALL,,NE1*2
LSEL,,LOC,X,B,B*1.1414
LCCAT,ALL
LESIZE,ALL,,NE1
CSYS,0
LSEL,,LOC,Y,B+0.1,L-0.1
LESIZE,ALL,,NE2
LSEL,,LOC,Y,L,L
LESIZE,ALL,,NE1

```

LINGUAGGIO PARAMETRICO/2

PARAMETRIZZAZIONE “MESH”




LINGUAGGIO PARAMETRICO/3

PARAMETRIZZAZIONE CARICO E MESHATURA

```
C*****
C*** RESET DATABASE
C*****
FINISH
/CLEAR
C***
C*** PARAMETRI
C***
L=100      ! Lunghezza
B=25       ! Larghezza
RI=12.5    ! Raggio foro
NE0=4      ! Numeri di elementi
NE1=5
NE2=8
PSI=-10
/PREP7
C***
C*** MODELLO SOLIDO
C***
RECTANGLE,0,B,0,B
CYL4,0,0,0,0,RI,360
ASBA,1,2
RECTNG,0,B,B,L
AGLUE,ALL
C***
C*** VINCOLI
LSEL,,LOC,Y,0,0
DL,ALL,,UY
LSEL,,LOC,X,B,B
DL,ALL,,UX
```

```
C***
C*** SUDDIVISIONE
C***
LSEL,,LOC,Y,0,0
LESIZE,ALL,,NE0
LSEL,,LOC,Y,RI,B
LSEL,R,LOC,X,0,0
LESIZE,ALL,,NE0
CSYS,1
LSEL,,LOC,X,RI,RI
LESIZE,ALL,,NE1*2
LSEL,,LOC,X,B,B*1.1414
LCCAT,ALL
LESIZE,ALL,,NE1
CSYS,0
LSEL,,LOC,Y,B+0.1,L-0.1
LESIZE,ALL,,NE2
LSEL,,LOC,Y,L,L
LESIZE,ALL,,NE1
C***
C*** CARICHI
C***
SFL,ALL,PRES,PSI
ALLS
C***
C*** MATERIALE
C***
MP,EX,1,210000
C***
C*** MESH
C***
ET,1,182
AMESH,ALL
```

```
C***
C*** SOLUZIONE
C***
/SOLU
SOLVE
C***
C*** POST-PROCESSING
C***
/POST1
PLNSOL,S,Y
```




LINGUAGGIO PARAMETRICO/4

INPUT INTERATTIVO PARAMETRI

```
C*****
C*** RESET DATABASE
C*****
FINISH
/CLEAR
C***
C*** PARAMETRI
C***
*ASK, L, Lunghezza :,100
*ASK, B, Larghezza :,25
*ASK, RI, Raggio foro:,12.5
*ASK, NE0, N° elementi lato 0:,4
*ASK, NE1, N° elementi lato 1:,5
*ASK, NE2, N° elementi lato 2:,8
*ASK, PSI, Carico:,-10
/PREP7
C***
C*** MODELLO SOLIDO
C***
RECTANGLE,0,B,0,B
CYL4,0,0,0,0,RI,360
ASBA,1,2
RECTNG,0,B,B,L
AGLUE,ALL
C***
C*** VINCOLI
LSEL,,LOC,Y,0,0
DL,ALL,,UY
LSEL,,LOC,X,B,B
DL,ALL,,UX
```

```
LSEL,,LOC,Y,0,0
LESIZE,ALL,,NE0
LSEL,,LOC,Y,RI,B
LSEL,R,LOC,X,0,0
LESIZE,ALL,,NE0
CSYS,1
LSEL,,LOC,X,RI,RI
LESIZE,ALL,,NE1*2
LSEL,,LOC,X,B,B*1.1414
LCCAT,ALL
LESIZE,ALL,,NE1
CSYS,0
LSEL,,LOC,Y,B+0.1,L-0.1
LESIZE,ALL,,NE2
LSEL,,LOC,Y,L,L
LESIZE,ALL,,NE1
SFL,ALL,PRES,PSI
ALLS
MP,EX,1,210000
ET,1,182
AMESH,ALL
```

***ASK, Var., Messaggio, "default"**



LINGUAGGIO PARAMETRICO/5

Istruzioni logiche:

***IF, EXPR. 1, OPER, EXPR. 2, THEN**

=

***ELSEIF, EXPR. 3, OPER, EXPR. 4**

=

***ELSE**

=

***ENDIF**

EQ, LE, GT, GE



Arresto esecuzione file di comandi:



/EOF

LINGUAGGIO PARAMETRICO/5

```

C*****
C*** RESET DATABASE
C*****
FINISH
/CLEAR
C***
C*** PARAMETRI
C***
*ASK, L, Lunghezza :,100
*ASK,NE0, N° elementi lato 0:,4
*ASK,PSI, Carico:,-10
C***
C*** CALCOLO PARAMETRI
C***
B=0.25*L
*IF,B,LE,10,THEN
      RI=0.25*B
*ELSEIF,B,LE,20
      RI=5
*ELSE
      RI=10
*ENDIF
NE1=NINT(NE0*B/(B-RI))
NE2=NINT(NE0*(L-B)/(B-RI))
/PREP7
C***
C*** MODELLO SOLIDO
C***
RECTANGLE,0,B,0,B
CYL4,0,0,0,0,RI,360
ASBA,1,2
RECTNG,0,B,B,L
AGLUE,ALL
C***
C*** VINCOLI
LSEL,,LOC,Y,0,0
DL,ALL,,UY
LSEL,,LOC,X,B,B
DL,ALL,,UX
C***
C*** MESH
C***
LSEL,,LOC,Y,0,0
LESIZE,ALL,,NE0
LSEL,,LOC,Y,RI,B
LSEL,R,LOC,X,0,0
LESIZE,ALL,,NE0
CSYS,1
LSEL,,LOC,X,RI,RI
LESIZE,ALL,,NE1*2
LSEL,,LOC,X,B,B*1.1414
LCCAT,ALL
LESIZE,ALL,,NE1
CSYS,0
LSEL,,LOC,Y,B+0.1,L-0.1
LESIZE,ALL,,NE2
LSEL,,LOC,Y,L,L
LESIZE,ALL,,NE1
SFL,ALL,PRES,PSI
ALLS
MP,EX,1,210000
ET,1,182
AMESH,ALL

```



LINGUAGGIO PARAMETRICO/6

Istruzioni di salto condizionato ed incondizionato:

***IF, Expr. 1, OPER, Expr. 2, :Label**

oppure

***GO, :Label**

=

:Label



Esempio:

*IF,A,LE,10,:LAB1

=

*GO,:LAB2

:LAB1

=

:LAB2

LINGUAGGIO PARAMETRICO/7

Istruzioni di esecuzione ciclica:

***DO, VAR, VAL. IN., VAL. FIN., PASSO**

=

=

***ENDDO**

ALTRI COMANDI UTILI

***GET**, *Par*, *Entity*, *ENTNUM*, *Item1*, *IT1NUM*, *Item2*, *IT2NUM*

Permette di richiedere qualunque informazione sul modello, inserendo la risposta nel parametro *Par*.

***GET**, *Par*, *KP*, *0*, *NUM*, *MAX*

Il numero d'ordine massimo dei KPs nel modello

***GET**, *Par*, *LINE*, *N*, *KP,1*

Il numero d'ordine massimo del KP 1 cui è collegata la linea N

ALTRI COMANDI UTILI

***STATUS**,*Par*

Restituisce il valore del parametro richiesto.

/OUTPUT,*fname,ext,,append*

Permette di registrare output di testo su file

/OUTPUT

Rimanda output su schermo

ALTRI COMANDI UTILI

***AFUN**, *Lab*

Permette di stabilire le unità di misura delle funzioni trigonometriche (SIN, COS, TAN, ASIN, ACOS, ATAN) in ingresso ed in uscita.

RAD --

Use radians for input and output of parameter angular functions (default).

DEG --

Use degrees for input and output of parameter angular functions.

STAT --

Show current setting (DEG or RAD) for this command

ALTRI COMANDI UTILI

\$

Permette di mettere più istruzioni sulla stessa riga in un file di comandi

```
CSYS,0 $LSEL,,LOC,Y,B+0.1,L-0.1  
LESIZE,ALL,,,NE2  
LSEL,,LOC,Y,L,L $LESIZE,ALL,,,NE1  
SFL,ALL,PRES,PSI
```